

6. SYMULACJA NIEUSTALONYCH PROCESÓW CIEPLNYCH

6.1. ZASADY OGÓLNE

Równanie Fouriera dla procesu niestacjonarnego (nieustalonego) ma postać:

$$\operatorname{div}(k(t)\operatorname{grad}(t)) + Q = c\rho \frac{\partial t}{\partial \tau},$$

albo w przypadku anizotropowych własności cieplnych:

$$\frac{\partial}{\partial x} \left(k_x(t) \frac{\partial t}{\partial x} \right) + \frac{\partial}{\partial y} \left(k_y(t) \frac{\partial t}{\partial y} \right) + \frac{\partial}{\partial z} \left(k_z(t) \frac{\partial t}{\partial z} \right) + \left(Q - c\rho \frac{\partial t}{\partial \tau} \right) = 0. \quad (6.1)$$

W określonej chwili czasu pochodne temperatury mogą być traktowane jako funkcje tylko współrzędnych x, y, z . Wtedy rozwiązanie równania (6.1) jest prowadzone analogicznie jak dla procesu ustalonego, przyjmując całe wyrażenie w ostatnim nawiasie (6.1) jako parametr Q w funkcjonale (5.9). W wyniku wykorzystania procedur, opisanych w rozdziale 5.1 otrzymamy:

$$[H]\{t\} + [C] \frac{\partial}{\partial \tau} \{t\} + \{P\} = 0, \quad (6.2)$$

$$[C] = \int_V c\rho \{N\} \{N\}^T dV. \quad (6.3)$$

W ogólnym przypadku wartości temperatury w węzłach $\{t\}$ zależą od czasu. Przyjmując, że wektor $\{t_0\}$ reprezentuje temperatury węzłowe w chwili $\tau = 0$, to w przedziale czasu $\Delta\tau$ wektor ten będzie wyznaczony równaniem:

$$\{t\} = \{N_0, N_1\} \begin{Bmatrix} \{t_0\} \\ \{t_1\} \end{Bmatrix}. \quad (6.4)$$

W równaniu (6.4) $\{N_0\}$ i $\{N_1\}$ są funkcjami kształtu zależnymi od czasu, $\{t_1\}$ – temperatury węzłowe po czasie $\Delta\tau$.

Zakładając, że dla małych kroków czasowych $\Delta\tau$ zależność temperatur węzłowych od czasu jest liniowa, funkcje kształtu przyjmą postać:

$$N_0 = \frac{\Delta\tau - \tau}{\Delta\tau}, \quad N_1 = \frac{\tau}{\Delta\tau}. \quad (6.5)$$

Uwzględniając zależność (6.4), pochodne temperatury względem czasu można przedstawić następująco:

$$\begin{aligned} \frac{\partial \{t\}}{\partial \tau} &= \begin{Bmatrix} \frac{\partial N_0}{\partial \tau} & \frac{\partial N_1}{\partial \tau} \end{Bmatrix} \begin{Bmatrix} \{t_0\} \\ \{t_1\} \end{Bmatrix} = \\ &= \frac{1}{\Delta\tau} \begin{Bmatrix} -1 & 1 \end{Bmatrix} \begin{Bmatrix} \{t_0\} \\ \{t_1\} \end{Bmatrix} = \frac{\{t_1\} - \{t_0\}}{\Delta\tau}. \end{aligned} \quad (6.6)$$

Ponieważ podczas interpolacji czasowej wykorzystano liniowe funkcje kształtu, rezultat (6.6) jest zbieżny z wynikiem, który można uzyskać za pomocą metody różnic skończonych. Istnieje kilka możliwości rozwiązania układu równań (6.2) za pomocą interpolacji (6.6) zależnie od tego, w jakiej chwili czasu będziemy rozpatrywać wektor $\{t\}$ we wzorze (6.2) przy macierze $[H]$.

1. Przyjęto, że $\{t\} = \{t_0\}$, dlatego wzór (6.2) można zapisać jako:

$$[H]\{t_0\} + [C]\frac{\{t_1\} - \{t_0\}}{\Delta\tau} + \{P\} = 0.$$

Na tej podstawie otrzymano jawny schemat wyznaczenia temperatury $\{t_1\}$ w przedziale czasu $\Delta\tau$:

$$\{t_1\} = \{t_0\} - \frac{\Delta\tau}{[C]}([H]\{t_0\} + \{P\}). \quad (6.7)$$

Jednak taki schemat ma ograniczone zastosowanie, ze względu na słabą stabilność rozwiązania dla różnych $\Delta\tau$.

2. Przyjęto, że $\{t\} = \{t_1\}$, dlatego wzór (6.2) można zapisać następująco:

$$[H]\{t_1\} + [C]\frac{\{t_1\} - \{t_0\}}{\Delta\tau} + \{P\} = 0.$$

W ten sposób otrzymano niejawny schemat wyznaczenia temperatury $\{t_1\}$:

$$\left([H] + \frac{[C]}{\Delta\tau}\right)\{t_1\} - \left(\frac{[C]}{\Delta\tau}\right)\{t_0\} + \{P\} = 0. \quad (6.8)$$

W takim przypadku wyznaczenie $\{t_1\}$ wymaga rozwiązania układu równań (6.8).

3. Kolejny możliwy schemat traktuje temperaturę $\{t\}$, która wchodzi do równania (6.2), jako średnią z temperatur w chwili $\tau = 0$ i w chwili $\tau = \Delta\tau$: $\{t\} = \frac{1}{2}(\{t_0\} + \{t_1\})$. W związku z powyższym koniecznym staje się rozwiązanie następującego układu równań:

$$\begin{aligned} \{P\}^* &= \frac{1}{2}(\{P_1\} + \{P_0\}), \\ [H]\frac{1}{2}(\{t_0\} + \{t_1\}) + [C]\frac{\{t_1\} - \{t_0\}}{\Delta\tau} + \{P\}^* &= 0. \end{aligned}$$

Ostatnie z równań można przekształcić w następujący sposób:

$$\begin{aligned} \left(\frac{[H]}{2} + \frac{[C]}{\Delta\tau}\right)\{t_1\} + \left(\frac{[H]}{2} - \frac{[C]}{\Delta\tau}\right)\{t_0\} + \{P\}^* &= 0, \\ \left([H] + \frac{2[C]}{\Delta\tau}\right)\{t_1\} + \left([H] - \frac{2[C]}{\Delta\tau}\right)\{t_0\} + 2\{P\}^* &= 0. \end{aligned} \quad (6.9)$$

4. Uwzględnienie czasu również możliwe jest za pomocą metody reszt ważonych. Ponieważ wektor temperatur węzłowych $\{t_0\}$ jest znany, w celu przeprowadzenia całkowania wyrażenia (6.6) względem czasu należy wprowadzić tylko jedną ważoną rezydualną $N_1 = \frac{\tau}{\Delta\tau}$ zgodnie z zależnością:

$$\int_0^{\Delta\tau} \frac{\tau}{\Delta\tau} \left([H]\{N_0, N_1\} \begin{Bmatrix} \{t_0\} \\ \{t_1\} \end{Bmatrix} + [C] \begin{Bmatrix} \frac{\partial N_0}{\partial \tau} \\ \frac{\partial N_1}{\partial \tau} \end{Bmatrix} \begin{Bmatrix} \{t_0\} \\ \{t_1\} \end{Bmatrix} + \{P\} \right) d\tau = 0. \quad (6.10)$$

Podstawiając funkcje kształtu N_0 i N_1 ze wzoru (6.5) do powyższego równania otrzymuje się:

$$\int_0^{\Delta\tau} \frac{\tau}{\Delta\tau} \left([H] \left(\frac{\Delta\tau - \tau}{\Delta\tau} \{t_0\} + \frac{\tau}{\Delta\tau} \{t_1\} \right) + \frac{[C]}{\Delta\tau} (-\{t_0\} + \{t_1\}) + \{P\} \right) d\tau = 0. \quad (6.11)$$

Następnie po przeprowadzeniu całkowania i przekształceniach otrzymuje się:

$$\left(2[H] + \frac{3}{\Delta\tau}[C]\right)\{t_1\} + \left([H] - \frac{3}{\Delta\tau}[C]\right)\{t_0\} + 3\{P\} = 0. \quad (6.12)$$

Wyrażenie (6.12) jest układem liniowych równań algebraicznych pozwalających obliczyć wartości temperatur węzłowych $\{t_1\}$ po czasie $\Delta\tau$ przy zadanych temperaturach $\{t_0\}$ w chwili $\tau = 0$.

Przedstawiony sposób w porównaniu z metodami 2 i 3 jest bardziej logiczny i stanowi ich uogólnienie. Na podstawie praktycznego doświadczenia Autora można stwierdzić, że w większości przypadków podejścia oparte na metodach 2-4 dają zbliżone wyniki.


6.2. WYZNACZANIE NIEUSTALONEGO ROZKŁADU TEMPERATURY WE WSADZIE O PRZEKROJU OKRĄGŁYM

Jako przykład problemu wyznaczenia niustalonego rozkładu temperatury rozpatrzmy proste jednowymiarowe zadanie. Analizowano proces niustalonego przewodnictwa ciepła we wsadzie o przekroju okrągłym (rys. 6.1). Założono, że wymiana ciepła będzie odbywała się w sposób osiowo-symetryczny (rys. 6.2).


Na powierzchni zewnętrznej kęsa zachodzi wymiana ciepła przez konwekcję. Współczynnik konwekcyjnej wymiany ciepła jest równy α , temperatura otoczenia – t_∞ . Omawiane zadanie rozpatrzono w cylindrycznym układzie współrzędnych (rys. 6.2).

Dla rozpatrywanego przypadku funkcjonal (5.9) można zapisać w następujący sposób:

$$J = \int_V \frac{k}{2} \left(\frac{dt}{dr}\right)^2 dV - \int_V Q t dV + \int_S \frac{\alpha}{2} (t - t_\infty)^2 dS \quad (6.13)$$



Rys. 6.1. Przykładowy rozkład temperatury we wsadzie o przekroju okrągłym



Rys. 6.2. Schemat obliczeniowy do wyznaczenia niustalonego rozkładu temperatury

Rozpatrzono proces minimalizacji funkcjonału (6.13) dla jednego wybranego elementu (rys. 6.2). Temperatura wewnątrz elementów jest zdefiniowana wzorem:

$$t = \{N_j^T\} \{t\} = N_i t_i + N_j t_j = \frac{r_j - r}{\Delta r} t_i + \frac{r - r_i}{\Delta r} t_j, \quad (6.14)$$

gdzie:

Δr – krok siatki, $\Delta r = r_j - r_i$;

N_i, N_j – funkcje kształtu;

t_i i t_j – temperatury w węzłach rozpatrywanego elementu po czasie $\Delta \tau$.

Wyznaczono pochodną temperatury względem r :

$$\frac{\partial t}{\partial r} = \frac{\partial N_i}{\partial r} t_i + \frac{\partial N_j}{\partial r} t_j. \quad (6.15)$$

Obliczono całki objętościowe funkcjonału (6.13). W tym celu wyznaczono parametry całkowania:

$$dV = 2\pi r L dr, \quad (6.16)$$

$$\int_S dS = 2\pi r_{\max} L, \quad (6.17)$$

gdzie:

L – długość wsadu;

r_{\max} – promień wsadu (rys. 6.2).

Obliczenie pierwszej całki we wzorze (6.13) sprowadza się do:

$$\begin{aligned} \int_V \frac{k}{2} \left(\frac{dt}{dx} \right)^2 dV &= \int_{r_i}^{r_j} \frac{k}{2} \left(\frac{\partial N_i}{\partial r} t_i + \frac{\partial N_j}{\partial r} t_j \right)^2 2\pi r L dr = \pi L \int_{r_i}^{r_j} k \left(\frac{t_j - t_i}{\Delta r} \right)^2 r dr = \\ &= \pi L k \left(\frac{t_j - t_i}{\Delta r} \right)^2 \sum_{p=1}^{n_p} (r_p w_p \det[J]), \end{aligned} \quad (6.18)$$

gdzie:

$\det[J]$ – wyznacznik macierzy Jacobiego, który jest Jakobianem transformacji układu współrzędnych $0r$ do układu 0ξ , w którym element ma długość 2 (rys. 6.3, a);

w – współczynniki wagi w punktach całkowania Gaussa r_p (rys. 6.3).

Należy zaznaczyć, że w rozpatrywanym przypadku można było analitycznie wyznaczyć całki we wzorze (6.15). Jednak dla bardziej skomplikowanych elementów nie jest to możliwe, więc wykorzystuje się metodę całkowania numerycznego, która jest bardziej charakterystyczna dla MES.


Dla omawianego przypadku (jednowymiarowego) transformacja układu współrzędnych (rys. 6.3) jest wyznaczona następującym wzorem:

$$r = N_i r_i + N_j r_j = \frac{1}{2}(1 - \xi)r_i + \frac{1}{2}(1 + \xi)r_j. \quad (6.19)$$

Wartość wyznacznika macierzy Jacobiego określona jest następującym wzorem:

$$\det[J] = \det \left[\frac{dr}{d\xi} \right] = \frac{\partial N_i}{\partial \xi} r_i + \frac{\partial N_j}{\partial \xi} r_j = \frac{r_j - r_i}{2} = \frac{\Delta r}{2}. \quad (6.20)$$

We wzorze (6.19) funkcje kształtu elementu są zapisane w układzie lokalnym (rys. 6.3).



Rys. 6.3. Wybrany element skończony w układach lokalnym 0ξ (a) i globalnym $0r$ (b)

Po wprowadzeniu równania (6.20) do wzoru (6.18) otrzymano:

$$\begin{aligned} \int_V \frac{k}{2} \left(\frac{dt}{dx} \right)^2 dV &= \pi L k \left(\frac{t_j - t_i}{\Delta r} \right)^2 \sum_{p=1}^{n_p} (r_p w_p \det J) = \\ &= \pi L k \frac{\Delta r}{2} \left(\frac{t_j - t_i}{\Delta r} \right)^2 \sum_{p=1}^{n_p} (r_p w_p) \end{aligned} \quad (6.21)$$

Za pomocą funkcji Q we wzorze (6.13) można uwzględnić niestacjonarność procesu zmiany temperatury, co zostanie opisane poniżej. Założono, że funkcja Q jest równa:

$$Q = c\rho \frac{dt}{d\tau} \quad (6.22)$$

Wtedy wariacyjne sformułowanie (6.13) jest równoważne sformułowaniu równania Fouriera (6.1). Po wprowadzeniu równań (6.20) i (6.22) do odpowiedniej części wzoru (6.13) otrzymano:

$$\begin{aligned} \int_V Q t dV &= \int_V c\rho \frac{dt}{d\tau} t dV = 2\pi L c\rho \int_{r_i}^{r_j} \frac{dt}{d\tau} t r dr = \\ &= \pi L c\rho \sum_{p=1}^{n_p} \left(\frac{dt}{d\tau} (N_i t_i + N_j t_j) r_p w_p \Delta r \right). \end{aligned} \quad (6.23)$$

Całka liczona po powierzchni zewnętrznej będącej w kontakcie z otoczeniem jest zapisana jako:

$$\int_S \frac{1}{2} \alpha (t - t_\infty)^2 dS = \pi L r_{\max} \alpha (t_j - t_\infty)^2 \quad (6.24)$$

Po podstawieniu wzorów (6.21), (6.23) oraz (6.24) do funkcjonału (6.13) dla wybranego elementu e otrzymamy:

$$\begin{aligned} J_e &= \pi L k \frac{\Delta r}{2} \left(\frac{t_j - t_i}{\Delta r} \right)^2 \sum_{p=1}^{n_p} (r_p w_p) + \\ &+ \pi L c\rho \sum_{p=1}^{n_p} \left(\frac{dt}{d\tau} (N_i t_i + N_j t_j) r_p w_p \Delta r \right) + \pi L r_{\max} \alpha (t_j - t_\infty)^2, \end{aligned}$$

lub po przekształceniu:

$$\begin{aligned} J_e &= k \frac{\Delta r}{2} \left(\frac{t_j - t_i}{\Delta r} \right)^2 \sum_{p=1}^{n_p} (r_p w_p) + \\ &+ c\rho \Delta r \sum_{p=1}^{n_p} \left(\frac{dt}{d\tau} (N_i t_i + N_j t_j) r_p w_p \right) + r_{\max} \alpha (t_j - t_\infty)^2. \end{aligned} \quad (6.25)$$

Dla minimalizacji funkcjonału (6.25) wykorzystano warunek ekstremum funkcji:

$$\left. \begin{aligned} \frac{\partial J_e}{\partial t_i} &= 0 \\ \frac{\partial J_e}{\partial t_j} &= 0 \end{aligned} \right\}. \quad (6.26)$$

Po zróżniczkowaniu równania (6.25) względem temperatury t_i w węzłach, otrzymano:

$$\begin{aligned} \frac{\partial J_e}{\partial t_i} &= k\Delta r \left(\frac{t_j - t_i}{\Delta r} \right) \left(-\frac{1}{\Delta r} \right) \sum_{p=1}^{n_p} r_p w_p + \\ &+ c\rho\Delta r \sum_{p=1}^{n_p} \left(\frac{dt}{d\tau} N_i r_p w_p \right) = 0, \\ \frac{dt}{d\tau} &= \frac{N_i t_i + N_j t_j - N_i t_{i0} - N_j t_{j0}}{\Delta \tau}, \end{aligned} \quad (6.27)$$

gdzie

t_{i0} i t_{j0} – temperatury w węzłach rozpatrywanego elementu w poprzedniej chwili czasu (w chwili $\tau = 0$);

$\Delta \tau$ – przyrost czasu.

Po przekształceniu równania (6.27) otrzymano:

$$\begin{aligned} \frac{\partial J_e}{\partial t_i} &= k\Delta r \left(\frac{t_j - t_i}{\Delta r} \right) \left(-\frac{1}{\Delta r} \right) \sum_{p=1}^{n_p} r_p w_p + \\ &+ \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} \left((N_i t_i + N_j t_j - N_i t_{i0} - N_j t_{j0}) N_i r_p w_p \right) = 0. \end{aligned}$$

Po przekształceniu tego równania otrzymano:


$$\begin{aligned} \frac{\partial J_e}{\partial t_i} &= t_i \left(\frac{k}{\Delta r} \sum_{p=1}^{n_p} r_p w_p + \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} N_i^2 r_p w_p \right) + \\ &+ t_j \left(-\frac{k}{\Delta r} \sum_{p=1}^{n_p} r_p w_p + \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} N_i N_j r_p w_p \right) - \\ &+ \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} (N_i t_{i0} + N_j t_{j0}) N_i r_p w_p = 0. \end{aligned} \quad (6.28)$$

Analogicznie dla t_j :

$$\begin{aligned} \frac{\partial J_e}{\partial t_j} &= t_i \left(-\frac{k}{\Delta r} \sum_{p=1}^{n_p} r_p w_p + \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} N_i N_j r_p w_p \right) + \\ &+ t_j \left(\frac{k}{\Delta r} \sum_{p=1}^{n_p} r_p w_p + \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} N_j^2 r_p w_p + 2\alpha r_{\max} \right) - \\ &+ \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} (N_i t_{i0} + N_j t_{j0}) N_j r_p w_p - 2\alpha r_{\max} t_{\infty} = 0. \end{aligned} \quad (6.29)$$

Równania (6.28) i (6.29) zapisano w postaci macierzowej:

$$[K_e] \{t\} + \{F_e\} = 0, \quad (6.30)$$



Rys. 6.4. Schemat podziału na elementy z globalną i lokalną numeracją węzłów

Dalej rozpatrzone rozwiązanie tego samego problemu w oparciu o rozwiązanie ogólne (6.8), które jest powtórzone poniżej:

$$\left([H] + \frac{[C]}{\Delta\tau} \right) \{t_1\} - \left(\frac{[C]}{\Delta\tau} \right) \{t_0\} + \{P\} = 0.$$

Dla omawianego przypadku wzory (5.14), (5.15) oraz równanie (6.3) można zapisać następująco (dla dowolnego elementu skończonego):

$$[C] = \int_V c\rho \{N\} \{N\}^T dV,$$

$$[H] = \int_V k \left\{ \frac{\partial \{N\}}{\partial r} \right\} \left\{ \frac{\partial \{N\}}{\partial r} \right\}^T dV + \int_S \alpha \{N\} \{N\}^T dS,$$

$$\{P\} = - \int_S \alpha \{N\} t_\infty dS,$$

gdzie analogicznie (6.16) i (6.17) mamy:

$$dV = 2\pi r L dr,$$

oraz:

$$\int_S dS = 2\pi r_{\max} L.$$

Całkowanie po promieniu r wykonano za pomocą całkowania numerycznego:

$$\int_{r_i}^{r_j} r dr = \sum_{p=1}^{n_p} r_p w_p \det[J] = \frac{\Delta r}{2} \sum_{p=1}^{n_p} (r_p w_p).$$

Wektor funkcji kształtu dla elementu jednowymiarowego pierwszego rzędu można zapisać następująco:

$$\{N\} = \begin{Bmatrix} N_i \\ N_j \end{Bmatrix} = \begin{Bmatrix} \frac{r_j - r}{\Delta r} \\ \frac{r - r_i}{\Delta r} \end{Bmatrix}; \quad \{N\}^T = \{N_i \quad N_j\} = \left\{ \frac{r_j - r}{\Delta r} \quad \frac{r - r_i}{\Delta r} \right\}. \quad (6.34)$$

Macierz pochodnych funkcji kształtu dla elementu jednowymiarowego można wyrazić w oparciu o równania (6.34) jako:

$$\left\{ \frac{\partial \{N\}}{\partial r} \right\} = \begin{Bmatrix} -\frac{1}{\Delta r} \\ \frac{1}{\Delta r} \end{Bmatrix}; \quad \left\{ \frac{\partial \{N\}}{\partial r} \right\}^T = \left\{ -\frac{1}{\Delta r} \quad \frac{1}{\Delta r} \right\}. \quad (6.35)$$

Zgodnie z otrzymanymi wzorami można zapisać macierz [C]:

$$\begin{aligned}
[C] &= \int_V c\rho\{N\}\{N\}^T dV = \int_V 2\pi L c\rho\{N\}\{N\}^T r dr = \\
&= 2\pi L c\rho \int_V \begin{Bmatrix} N_i \\ N_j \end{Bmatrix} \begin{Bmatrix} N_i & N_j \end{Bmatrix} r dr = \\
&= \pi L c\rho \Delta r \sum_{p=1}^{n_p} \left(r_p w_p \begin{bmatrix} N_i N_i & N_i N_j \\ N_i N_j & N_j N_j \end{bmatrix} \right)
\end{aligned} \tag{6.36}$$

oraz macierz [H]:

$$\begin{aligned}
[H] &= \int_V k \left\{ \frac{\partial\{N\}}{\partial r} \right\} \left\{ \frac{\partial\{N\}}{\partial r} \right\}^T dV + \int_S \alpha\{N\}\{N\}^T dS = \\
&= \int_V 2\pi L k \begin{Bmatrix} -\frac{1}{\Delta r} \\ \frac{1}{\Delta r} \end{Bmatrix} \begin{Bmatrix} -\frac{1}{\Delta r} & \frac{1}{\Delta r} \end{Bmatrix} r dr + 2\pi L \alpha \begin{bmatrix} N_i N_i & N_i N_j \\ N_i N_j & N_j N_j \end{bmatrix} r_{\max} = \\
&= 2\pi L k \begin{bmatrix} \frac{1}{\Delta r^2} & -\frac{1}{\Delta r^2} \\ -\frac{1}{\Delta r^2} & \frac{1}{\Delta r^2} \end{bmatrix} \int_{r_i}^{r_j} r dr + 2\pi L \alpha \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix} r_{\max} = \\
&= \pi L k \begin{bmatrix} \frac{1}{\Delta r} & -\frac{1}{\Delta r} \\ -\frac{1}{\Delta r} & \frac{1}{\Delta r} \end{bmatrix} \sum_{p=1}^{n_p} (r_p w_p) + 2\pi L \alpha \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix} r_{\max}
\end{aligned} \tag{6.37}$$

i wektor obciążeń {P}:

$$\begin{aligned}
\{P\} &= - \int_S \alpha \begin{Bmatrix} N_i \\ N_j \end{Bmatrix} t_{\infty} dS = -2\pi L r_{\max} t_{\infty} \alpha \begin{Bmatrix} N_i \\ N_j \end{Bmatrix} = \\
&= -2\pi L r_{\max} t_{\infty} \alpha \begin{Bmatrix} 0 \\ 1 \end{Bmatrix}.
\end{aligned} \tag{6.38}$$

Po podstawieniu otrzymanych wzorów do równania (6.8) uzyskano równanie macierzowe:

$$\begin{aligned}
&\left(\pi L k \begin{bmatrix} \frac{1}{\Delta r} & -\frac{1}{\Delta r} \\ -\frac{1}{\Delta r} & \frac{1}{\Delta r} \end{bmatrix} \sum_{p=1}^{n_p} (r_p w_p) + 2\pi L \alpha \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix} r_{\max} + \frac{\pi L c\rho \Delta r}{\Delta \tau} \sum_{p=1}^{n_p} \left(r_p w_p \begin{bmatrix} N_i N_i & N_i N_j \\ N_i N_j & N_j N_j \end{bmatrix} \right) \right) \{t_1\} + \\
&- \left(\frac{\pi L c\rho \Delta r}{\Delta \tau} \sum_{p=1}^{n_p} \left(r_p w_p \begin{bmatrix} N_i N_i & N_i N_j \\ N_i N_j & N_j N_j \end{bmatrix} \right) \right) \{t_0\} - 2\pi L r_{\max} t_{\infty} \alpha \begin{Bmatrix} 0 \\ 1 \end{Bmatrix} = 0,
\end{aligned}$$

które po uproszczeniu można zapisać następująco:

$$\begin{aligned}
&\left(\frac{k}{\Delta r} \begin{bmatrix} 1 & -1 \\ -1 & 1 \end{bmatrix} \sum_{p=1}^{n_p} (r_p w_p) + 2\alpha \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix} r_{\max} + \frac{\rho c \Delta r}{\Delta \tau} \sum_{p=1}^{n_p} \left(r_p w_p \begin{bmatrix} N_i N_i & N_i N_j \\ N_i N_j & N_j N_j \end{bmatrix} \right) \right) \{t_1\} \\
&- \left(\frac{\rho c \Delta r}{\Delta \tau} \sum_{p=1}^{n_p} \left(r_p w_p \begin{bmatrix} N_i N_i & N_i N_j \\ N_i N_j & N_j N_j \end{bmatrix} \right) \right) \{t_0\} - 2r_{\max} t_{\infty} \alpha \begin{Bmatrix} 0 \\ 1 \end{Bmatrix} = 0.
\end{aligned} \tag{6.39}$$

Równanie (6.39) można zapisać jako:

$$[K]\{t\} + \{F\} = 0,$$

albo:

$$\begin{bmatrix} K_{11} & K_{12} \\ K_{21} & K_{22} \end{bmatrix} \begin{Bmatrix} t_i \\ t_j \end{Bmatrix} + \begin{Bmatrix} F_1 \\ F_2 \end{Bmatrix} = 0,$$

gdzie składowe macierzy $[K]$ oraz wektora $\{F\}$ są równe:

$$\begin{aligned} K_{11} &= \frac{k}{\Delta r} \sum_{p=1}^{n_p} r_p w_p + \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} N_i^2 r_p w_p \\ K_{12} &= -\frac{k}{\Delta r} \sum_{p=1}^{n_p} r_p w_p + \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} N_i N_j r_p w_p \\ K_{21} &= -\frac{k}{\Delta r} \sum_{p=1}^{n_p} r_p w_p + \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} N_i N_j r_p w_p \\ K_{22} &= \frac{k}{\Delta r} \sum_{p=1}^{n_p} r_p w_p + \frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} N_j^2 r_p w_p + 2\alpha r_{\max} \\ F_1 &= -\frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} (N_i t_{i0} + N_j t_{j0}) N_i r_p w_p \\ F_2 &= -\frac{c\rho\Delta r}{\Delta \tau} \sum_{p=1}^{n_p} (N_i t_{i0} + N_j t_{j0}) N_j r_p w_p - 2\alpha r_{\max} t_{\infty}. \end{aligned} \tag{6.40}$$

Z tego wynika, że otrzymano taki sam wynik, jak przedstawiony w równaniach (6.31).

6.4 PRZYKŁAD OPRACOWANIA OPROGRAMOWANIA I OBLICZEŃ

Omówione w poprzednim rozdziale zadanie jest przedstawione poniżej w postaci oprogramowania TEMP1d napisanego w języku FORTRAN 90. Program główny służy do wczytywania danych do obliczeń z pliku DataInpTemp1d.txt. Wczytywane są następujące dane:

- Rmin - promień minimalny wsadu, m;
- Rmax - promień maksymalny wsadu, m;
- AlfaAir - współczynnik konwekcyjnej wymiany ciepła ($W/m^2 \text{ } ^\circ C$);
- TempBegin - temperatura początkowa, $^\circ C$;
- TempAir - temperatura otoczenia, $^\circ C$;
- TauMax - czas procesu, s;
- C - efektywne ciepło właściwe, $J/(kg^\circ C)$;
- K - współczynnik przewodzenia ciepła, $W/(m^\circ C)$;
- Ro - gęstość metalu, kg/m^4 .

Dane do obliczeń testowych z pliku DataInpTemp1d.txt są przedstawione na rys. 6.5.

```
*****
0.0      Rmin,      m
0.05    Rmax ,     m
300     AlfaAir,   W/m2 C;
100     TempBegin  °C;
1200    TempAir    °C;
700     C          J/(kg*C);
```

```

7800      Ro          kg/(m3);
25        K          W/(mC)
1000     TauMax      s;
*****

```

Rys. 6.5. Przykładowa zawartość pliku DataInpTemp1d.txt.

Główne obliczenia są wykonane w podprogramie **Temp1d** (rys. 6.6). Ten podprogram wywołuje podprogram **DLSLTR** z biblioteki numerycznej [25] FORTRANu, który rozwiązuje układ równań liniowych metodą eliminacji Gaussa. Opis najważniejszych zmiennych jest podany w kodzie, natomiast są też wykorzystywane oznaczenia, które pokazano na rys. 4.26.

Po wykonaniu obliczeń wyniki są zapisywane w pliku temperat.txt.

Wyniki obliczeń dla omówionego zadania pokazano na rys. 6.7.

Zaprezentowany program może zostać wykorzystany do symulacji procesów wymiany ciepła przy założeniu warunków jednowymiarowej wymiany ciepła (nagrzewanie albo chłodzenie wsadu o przekroju okrągłym).

```

program TEMP1D
|*****
! PROGRAM: TEMP1D
! Autor: Milenin Andriy, CopyRight, 2006-2008
! Akademia Górniczo-Hutnicza, milenin@agh.edu.pl
|*****
! Dane wejściowe
! Rmin - promień minimalny, m
! Rmax - promień maksymalny, m
! AlfaAir - współczynnik konwekcyjnej wymiany ciepła (W/m2 C)
! TempBegin - temperatura początkowa, C
! TempAir - temperatura środowiska, C
! TauMax - czas procesu, s
! C - efektywne ciepło właściwe, J/(kg*C)
! K - współczynnik przewodzenia ciepła, W/(mC)
! Ro - gęstość metalu, kg/(m3)
|*****
implicit none;
real*8   :: Rmin,Rmax,AlfaAir,TempBegin,TempAir,TauMax;
real*8   :: t1, t2, Tau1, Tau2;
real*8   :: C,Ro,K;
integer*4 :: nRead;

nRead=217;
OPEN (nRead,FILE='DataInpTemp1d.txt')
READ(nRead,*)
READ(nRead,*) Rmin;
READ(nRead,*) Rmax;
READ(nRead,*) AlfaAir;
READ(nRead,*) TempBegin;
READ(nRead,*) t1, t2;
READ(nRead,*) C;
READ(nRead,*) Ro;
READ(nRead,*) K;
READ(nRead,*) Tau1, Tau2 ;
CLOSE (nRead);
  CALL Temp_1d( Rmin,Rmax,AlfaAir,TempBegin,t1,t2,C,Ro,K,Tau1, Tau2 );
end program TEMP1D;

SUBROUTINE Temp_1d( Rmin,Rmax,AlfaAir,TempBegin, t1,t2, C,Ro,K, Tau1, Tau2 );
use msimsl;
implicit none;
real*8   :: Rmin,Rmax,AlfaAir,TempBegin,TempAir,C,Ro,K,TauMax;
integer*4 :: nh, ne;
real*8   :: dTau,Tau, x,dR,a,Alfa;
real*8   :: t1,t2, Tau1, Tau2, dTmax, dT;
real*8   :: E(2),W(2),N1(2),N2(2),r(2),Ke(2,2),Fe(2),TempTau(2);
integer*4 :: i,nTau,ie,ip,Np,iTime;

```

```

real*8    :: Rp, TpTau;
integer*4  :: nNe, nTime;
integer*4  :: nPrint;
real*8, dimension(:), pointer :: vrtxTemp;
real*8, dimension(:), pointer :: vrtxCoordX;
real*8, dimension(:), pointer :: aC, aD, aE, aB;

nPrint=314;
OPEN (nPrint, FILE='temperat.txt')
WRITE(nPrint, (" Time,s      tc      tpov"));
WRITE(nPrint, ("*****"));

nh = 51;
ne = nh-1;
Np = 2;
a = K / (C*Ro);
W(1)=1;
W(2)=1;
E(1)=-0.5773502692;
E(2)= 0.5773502692;
N1(1) = 0.5*( 1-E(1) );
N1(2) = 0.5*( 1-E(2) );
N2(1) = 0.5*( 1+E(1) );
N2(2) = 0.5*( 1+E(2) );
dR = (Rmax-Rmin)/ne;
dTau = (dR**2)/(0.5*a);
TauMax = Tau1 + Tau2;
nTime=(TauMax/dTau) + 1;
dTau = TauMax / nTime;
ALLOCATE( vrtxTemp(nh) );
ALLOCATE( vrtxCoordX(nh) );
ALLOCATE( aC(nh) );
ALLOCATE( aD(nh) );
ALLOCATE( aE(nh) );
ALLOCATE( aB(nh) );
! Rozbudowa siatki elementów skończonych
x=0;
do i=1,nh
vrtxCoordX(i) = x;
vrtxTemp(i) = TempBegin;
x = x + dR;
end do;

dTmax = 0;
Tau=0;
do iTime = 1, nTime
aC=0;
aD=0;
aE=0;
aB=0;

TempAir = t1;
if(Tau>=Tau1) then
TempAir = t2;
end if;

do ie = 1, ne
r(1) = vrtxCoordX(ie);
r(2) = vrtxCoordX(ie+1);
TempTau(1) = vrtxTemp(ie);
TempTau(2) = vrtxTemp(ie+1);
dR = r(2)-r(1);

Alfa=0;
if (ie==ne) Alfa = AlfaAir;

Ke = 0;


```

```

Fe = 0;
do ip=1, Np
Rp = N1(ip)*r(1) + N2(ip)*r(2);
TpTau = N1(ip)*TempTau(1) + N2(ip)*TempTau(2);
!*****
! Obliczenie macierzy lokalnej
!*****
Ke(1,1) = Ke(1,1) + K*Rp*W(ip)/dR + C*Ro*dR*Rp*W(ip)*N1(ip)**2 /dTau;
Ke(1,2) = Ke(1,2) - K*Rp*W(ip)/dR + C*Ro*dR*Rp*W(ip)*N1(ip)*N2(ip)/dTau;
Ke(2,1) = Ke(1,2);
Ke(2,2) = Ke(2,2) + K*Rp*W(ip)/dR + C*Ro*dR*Rp*W(ip)*N2(ip)**2 /dTau + 2*Alfa*Rmax;
Fe(1) = Fe(1) + C*Ro*dR*TpTau*Rp*W(ip)*N1(ip)/dTau;
Fe(2) = Fe(2) + C*Ro*dR*TpTau*Rp*W(ip)*N2(ip)/dTau + 2*Alfa*Rmax*TempAir;
!*****
end do;
aD(ie) = aD(ie) + Ke(1,1);
aD(ie+1) = aD(ie+1) + Ke(2,2);
aE(ie) = aE(ie) + Ke(1,2);
aC(ie+1) = aC(ie+1) + Ke(2,1);
aB(ie) = aB(ie) + Fe(1);
aB(ie+1) = aB(ie+1) + Fe(2);
end do;
CALL DLSLTR (nh, aC, aD, aE, aB)
! Program rozwiązania układu równan z biblioteki FORTRAN [11].
do i=1,nh
vrtxTemp(i) = aB(i);
end do;
dT = dabs(vrtxTemp(1)-vrtxTemp(nh));
if (dT>dTmax) dTmax=dT;
WRITE(nPrint,(" ",4(" ",F12.2))) Tau,vrtxTemp(1),vrtxTemp(nh),dT;
Tau = Tau + dTau;
end do;
WRITE(nPrint,(" *****" ));
WRITE(nPrint,(" dTmax=",F12.2)) dTmax;
close (nPrint)
DEALLOCATE( vrtxTemp );
DEALLOCATE( vrtxCoordX );
DEALLOCATE( aC );
DEALLOCATE( aD );
DEALLOCATE( aE );
DEALLOCATE( aB );
END SUBROUTINE Temp_1d;

```

Rys. 6.6. Kod źródłowy programu TEMP1d



Rys. 6.7. Wyniki obliczenia nagrzewania wsadu o przekroju okrągłym, 1 – na powierzchni wsadu, 2 – w osi wsadu

Po dodaniu macierzy lokalnych wszystkich elementów skończonych do macierzy globalnej otrzymujemy macierz w postaci przedstawionej w tabeli 6.4.

Tabela 6.4.
Macierz globalna

	1	2	3	4	5	6	7	8	9
1	26,82	-23,18							
2	-23,18	114,56	-69,54						
3		-69,54	229,12	-115,90					
4			-115,90	343,68	-162,26				
5				-162,26	458,24	-208,62			
6					-208,62	572,80	-254,98		
7						-254,98	687,36	-301,34	
8							-301,34	801,92	-347,70
9								-347,70	527,42

Jak wynika z analizy tabeli 6.4 w macierzy globalnej niezerowe wyrazy są położone tylko w pobliżu przekątnej. Biorąc dodatkowo pod uwagę, że macierz ta jest symetryczna, można ją zastąpić macierzą prostokątną, której liczba kolumn jest równa szerokości pasma wyrazów niezerowych. Ta operacja pozwala na znaczne zmniejszenie pamięci komputera użytej do obliczeń.